

# 微机串行通信

## 微机串行通信

### 串行通信基础

#### 串行通信类型

#### 数据的传输方式

#### 串行异步通信协议

### 可编程串行异步 - 8250

#### 内部结构

#### 8250 初始化编程

### 串行通信程序设计

#### 串行通信外部环境

#### BIOS通信功能

#### 串口程序设计实例

## 串行通信基础

### 串行通信类型

- 串行通信：数据逐位顺序传送
  - 串行异步通信：一帧字符用起始位和停止位完成收发同步
  - 串行同步通信：靠同步字符来完成收发同步
- 并行通信：数据所有位同时被传输
- 数据传输速度：并行>串行同步>串行异步

### 数据的传输方式


- 单工方式




- 半双工方式



- 全双工方式



## 串行异步通信协议



- 一帧数据格式

一帧数据包括起始位、数据位、奇偶校验位、停止位

- 起始位

无数据传输则为1，有数据传输为0，检测到0开始接受数据

- 数据位

数据位个数5~8位，低位先传送，读信号反向读取

- 校验位

双方约定为检验逻辑1的奇校验或者偶校验，然后通过该位进行检验，也可以选择无校验位

- 停止位


可以是1、1.5、2位的逻辑1信号

由此我们可以得到1帧数据的范围为7~12位

- 通信速率

- 数据信号速率：单位时间传送信号的个数，单位为波特（B/S），称波特率、信号速率

- 数据通信速率：单位时间传送二进制比特的个数，单位为比特/秒（bps, b/s），称比特率或传信速率
- 我们采用二进制信号传输时，波特率=比特率，一位二进制为1波特
- 例如：传送一个字符“E”，（ASCII码为：1000101B =45H）7个数据位,奇校验, 2个停止位.



- RS-232C 串行接口标准


- TXD：发送数据（transmit Data）
- RXD：接受数据（Receive Data）
- RTS反：请求发送（Request to send）
- CTS反：允许发送（Clear to Send）
- DSR反：数据设备准备好（Data set Ready）
- DTR反：数据终端准备好（Data Term Ready）
- RS-232采用负逻辑，逻辑1为-15~-3V，逻辑0为+3V~15V
- 学会绘制TTL电平信号以及负逻辑的232的电平信号

## 可编程串行异步 - 8250

- 功能

- 实现数据串、并转换，实现全双工异步通信
- 支持异步通信协议，数据格式、通信速率由初始化编程设定
- 内部有中断机制、有MODEM控制器

- 内部结构



- 端口记忆

- 发送保持寄存器 (3F8H/2F8H)
- 接收缓存寄存器 (3F8H/2F8H)
- 通信线寄存器 (3FDH/2FDH)
- 中断允许寄存器 (3F9H/2F9H)
- 中断识别寄存器 (3FAH/2FAH)
- MODEM控制寄存器 (3FCH/2FCH)
- 除数寄存器 (高八位 3F9H/2F9H, 低八位 3F8H/2F8H)
- 通信线寄存器 (3FBH\2FBH)
- MODEM状态寄存器 (3FEH/2FEH)

- 程序员想要8250之间采用中断方式交换信息

- 中断允许寄存器相应位置1
- 8259的相应中断屏蔽位置0(主8259 IR3, IR4)
- CPU处于开中断
- MODEM控制寄存器D3=1, 即 $OUT_2$ 反=0, 打开中断请求通道

## 内部结构

- 除数寄存器（高八位 3F9H/2F9H，低八位 3F8H/2F8H）

存放分频系数

基准时钟：Fosc=1.8432MHz，发送器时钟：Ftxc=16\*通信速率

计算公式：


$$F_{osc}/\text{分频系数}=16*\text{通信速率}$$

其中16被称为波特率因子，通常可取 1、16、64

$$\text{分频系数} = \frac{1.8432\text{MHZ}}{16 * \text{通信速率}}$$

波特率	除数高 8 位	除数低 8 位	波特率	除数高 8 位	除数低 8 位
50	09H	00H	1800	00H	40H
75	06H	00H	2000	00H	3AH
110	04H	17H	2400	00H	30H
150	03H	00H	3600	00H	20H
300	01H	80H	4800	00H	18H
600	00H	C0H	7200	00H	10H
1200	00H	60H	9600	00H	0CH

- 通信线控制寄存器 - （3FBH\2FBH）



用于指定串行异步通信的数据格式，可以用来分辨数据是发送给发送保持寄存器或者分频系数

- 查收程序

```

SCAN:    MOV DX,3FDH
          IN AL,DX
          TEST AL,00000001B
          JZ SCAN
          MOV DX,3F8H
          IN AL,DX

DIS:      AND AL,7FH

```

- 查发程序'

```

SCAN:    MOV DX,3FDH
          IN AL,DX
          TEST AL,00100000B
          JZ SCAN
          MOV AL,
          MOV DX,3F8H
          OUT DX,AL

```

## 8250 初始化编程

- 确定波特率（设置除数寄存器）
- 确定数据格式（设置通信线路控制寄存器）
- 若使用中断方式（设置中断允许寄存器）
- 设置MODEM控制寄存器
  - 中断方式D3=1，允许8250送出中断信号
  - 查询方式D3=0
  - 内环自检D4=1
  - 正常通信D4=0
- 例. 编写子程序对主串口进行初始化
  - 通信速率=1200波特，一帧数据包括：8个数据位，1个停止位，无校验
  - 查询方式，完成内环自检

### 分析

- 1200波特 ——>0060H
- 一帧数据格式：00000011B = 03H
- 中断允许命令字 = 0
- MODEM控制字00010000H = 10H

### ;初始化子程序

```
I8250 PROC
```

```
    MOV AL,80H
```


```
    MOV DX,3FBH
```

```
    OUT TX,AL          ;设置波特率
```

```
MOV AL,00H
MOV DX,3F9H
OUT DX,AL
MOV AL,30H
MOV DX,3F8H
OUT DX,AL          ;设置确定数据格式
MOV DX,3FBH
MOV AL,00000011B   ;MOV AL,03H
OUT DX,AL          ;设置分频系数
MOV DX,3F9H
MOV AL,0
OUT DX,AL          ;设置中断允许寄存器
MOV DX,3FCH
MOV AL,10H
OUT DX,AL          ;设置MODEM控制寄存器
RET
I8250 ENDP
```

## 串行通信程序设计

### 串行通信外部环境



## BIOS通信功能

INT 14H向用户提供了4个中断子程序：串口初始化、发送一帧数据、接受一帧数据和测试通信线状态，初始化之后，内部中断会被禁止

- 串口初始化 —— 14H 0号功能
  - 入口参数：
    - AH=0串口初始化
    - AL=初始化参数
    - DX=0，对主串口初始化
    - DX=1，对辅串口初始化
  - 出口参数：
    - AH=通信线状态寄存器内容
    - AL=MODEM状态寄存状态
  - 初始化参数格式
    - D7、D6、D5参数来选择波特率
    - D4、D3用来校验选择
    - D2选择停止位长度
    - D1D0选择数据位长度
  - 功能执行步骤
    - ① 截取AL7~AL5查表,取出相应的波特率除数→除数寄存器。
    - ② 截取AL4~AL0→通信线控制寄存器。




- ③ 0→中断允许寄存器。
  - ④ 取通信线状态寄存器内容→AH。
  - ⑤ 取MODEM状态寄存器内容→AL。
  - ⑥ 执行IRET返回。
- 发送一帧数据 —— 14H 1号功能
  - 接受一帧数据 —— 14H 2号功能
  - 测试通信线状态 —— 14H 3号功能

功能	入口参数	出口参数
AH=1（发送数据）	AL = 欲发送字符	AH = 通信线路状态
	DX = 串行口号(0-2)	(AH)7 =1 表示传送失败
AH=2（接受数据）	DX = 串行口号(0-2)	AH = 通信线路状态
		(AH)7 =1 表示传送失败
		AL = 接收到的字符
AH=3（测试通信线状态）	DX = 串行口号（0-2）	AH = 通信线路状态
		AL = MODEM状态

串口程序设计实例

- 要求甲乙两台微机之间通过RS-232C接口进行短距离的串行通信，甲机作为发送端从主串口将一串字符逐个发送，乙机作为接收端在辅串口进行字符接收并显示在屏幕上，其中甲机发送字符串时以‘ETX’字符（ASCII码：03H）作为输入结束标志字符。通信双方约定波特率为2400波特，数据位7位，停止位1位，奇校验，发送和接收均采用查询方式。
- 通信示意图



- 设计思路

- 1. 甲乙两台微机之间进行短距离的单工通信，因此采用直接访问8250端口寄存器的方式进行编程，程序运行前将甲机的主串口与乙机的辅串口进行点到点单工通信连接
  - 2. 一串字符经逐个发送，发送字符和接收字符均采用查询方式。发送前，先读取通信线状态寄存器，查询发送保持寄存器是否空；接收前先读取通信线状态寄存器，查询一帧数据是否收完。
  - 3. 通信速率2400波特，分频系数为0030H，一帧字符有7个数据位，1个停止位，奇校验，数据帧格式字为0AH。
- 甲机发送程序

#### ;甲机发送程序

```

;FILENAME: 9_1.asm
.586
DATA    SEGMENT USE16
    BUF    DB        'Hello',03H
    LENS    EQU    $-BUF
DATA    ENDS
CODE    SEGMENT USE16
    ASSUME    CS:CODE, DS:DATA
BEG:     MOV        AX,DATA
            MOV        DS,AX
            CALL        I8250        ;主串口初始化
            LEA        BX,BUF
            MOV        CX,LENS
SCAN:    MOV        DX,3FDH
IN        AL,DX
            TEST        AL,20H        ;发送保持寄存器是否为空?
            JZ        SCAN
            MOV        DX,3F8H
            MOV        AL,[BX]        ;取字符
            OUT        DX,AL        ;送主串口数据寄存器
            INC        BX
            LOOP        SCAN
            MOV        AH,4CH
            INT        21H
I8250    PROC
    MOV        DX,3FBH
            MOV        AL,80H
            OUT        DX,AL        ;寻址位置1
            MOV        DX,3F9H
            MOV        AL,0
            OUT        DX,AL        ;写除数寄存器高8位
    MOV        DX,3F8H
            MOV        AL,30H
            OUT        DX,AL        ;写除数寄存器低8位
            MOV        DX,3FBH
            MOV        AL,0AH
            OUT        DX,AL        ;写数据帧格式
            MOV        DX,3F9H
            MOV        AL,0
            OUT        DX,AL        ;禁止8250内部中断
            MOV        DX,3FCH

```

```

                MOV     AL,0
                OUT     DX,AL ;8250正常收/发方式, 禁止中断
                RET
I8250          ENDP
CODE          ENDS
END            BEG

```

- 乙机接受程序

#### ;乙机接收程序

```

                ;FILENAME: 9_2.asm
                .586
DATA          SEGMENT    USE16
MSG           DB         'The received characters are', '$'
DATA          ENDS
CODE          SEGMENT    USE16
                ASSUME    CS:CODE,DS:DATA
BEG:          MOV     AX,DATA
                MOV     DS,AX
                CALL    I8250      ;辅串口初始化
                MOV     AH,9
                MOV     DX,OFFSET MSG      ;输出提示信息
                INT     21H
SCAN:        MOV     DX,2FDH
IN           AL,DX
                TEST    AL,01H            ;一帧数据收完否?
                JZ      SCAN
                MOV     DX,2F8H
                IN       AL,DX            ;读数据寄存器
                AND     AL,01111111B
                CMP     AL,03H            ;是否是结束标志字符
                JE      EXIT
                MOV     DL,AL
                MOV     AH,02H
                INT     21H
                JMP     SCAN
EXIT:        MOV     AH,4CH
                INT     21H
I8250        PROC
                MOV     DX,2FBH
                MOV     AL,80H
                OUT     DX,AL            ;寻址位置1
                MOV     DX,2F9H
                MOV     AL,0
                OUT     DX,AL            ;写除数寄存器高8位
                MOV     DX,2F8H
                MOV     AL,30H
                OUT     DX,AL            ;写除数寄存器低8位
                MOV     DX,2FBH
                MOV     AL,0AH
                OUT     DX,AL            ;写数据帧格式

```

```

MOV     DX,2F9H
MOV     AL,0
OUT     DX,AL      ;禁止8250内部中断
MOV     DX,2FCH
MOV     AL,0
OUT     DX,AL      ;8250正常收/发方式,禁止中断
RET
I8250   ENDP
CODE    ENDS
END      BEG

```


- 【例】 对主串口进行外环自动测试,将下列测试电文10行,经主串口发出,通过外环短路线接收,显示在屏幕上,测试电文如下:

THE QUICK BROWN FOX JUMPS OVER LAZY DOG

- 设计思路
  - 1. 在一条报路上,长时间的循环发送这条电文,接收方统计在一定时间内的差错率即可知道该报路的通信质量。
  - 2. 电文必须逐个字符发送,为了简化程序设计,发送字符和接收字符均采用查询方式,发送前,先读取通信线状态寄存器,查询发送保持寄存器空否? 接收前先读取通信线状态寄存器,查询一帧数据收完否?
  - 3. 本例采用两种方法编程:

1031\_1.ASM,直接访问8250端口寄存器,程序运行前RS232连接器按图(a)接线,没有使用联络线。

1031\_2.ASM,调用BIOS通信软件,程序运行前需按图(b)接线,准备好自发自收的外部环境。



程序1

```

;FILENAME:931_1.ASM
DATA    SEGMENT
TEXT    DB      'THE QUICK BROWN FOX JUMPS OVER LAZY DOG'
        DB      0DH,0AH

```

```

LLL      EQU      $-TEXT
ERROR    DB        'COM1 BAD !',0DH,0AH,'$'
DATA     ENDS
CODE     SEGMENT

        ASSUME    CS:CODE,DS:DATA
BEG:     MOV      AX,DATA
        MOV      DS,AX
        CALL     I8250                ;主串口初始化
        MOV      CH,10                ;10行送CH
AGAIN:   MOV      CL,LLL                ;电文长度送CL
        MOV      BX,OFFSET TEXT
TSCAN:   MOV      DX,3FDH
        IN       AL,DX
        TEST     AL,20H                ;发送保持寄存器空 ?
        JZ       TSCAN                ;否
        MOV      AL,[BX]                ;取字符
SEND:    MOV      DX,3F8H
        OUT      DX,AL                ;送主串口数据寄存器
        MOV      SI,0
RSCAN:   MOV      DX,3FDH
        IN       AL,DX
        TEST     AL,01H                ;一帧数据收完否 ?
        JNZ      REVEICE                ;收完转
        DEC      SI
        JNZ      RSCAN                ; 延时
        JMP      DISPERR                ;超时,转出错处理
REVEICE: MOV      DX,3F8H
        IN       AL,DX                ;读数据寄存器
        AND      AL,7FH
DISP:    MOV      AH,2
        MOV      DL,AL
        INT      21H                ;屏幕显示
        INC      BX
        DEC      CL                ;计数
        JNZ      TSCAN
        DEC      CH                ;行计数
        JNZ      AGAIN
        JMP      RETURN
DISPERR: MOV      AH,9
        MOV      DX,OFFSET ERROR
        INT      21H                ;显示出错信息
RETURN:  MOV      AH,4CH
        INT      21H                ;返回 DOS

;-----
I8250    PROC                                ;主串口初始化子程序
        MOV      DX,3FBH
        MOV      AL,80H
        OUT      DX,AL                ;寻址位置1
        MOV      DX,3F9H
        MOV      AL,00H
        OUT      DX,AL                ;写除数高8位
        MOV      DX,3F8H

```

```

MOV     AL,60H
OUT     DX,AL           ;写除数低8位
MOV     DX,3FBH
MOV     AL,03H
OUT     DX,AL           ;无校验传送,8位数据
MOV     DX,3F9H
MOV     AL,00H
OUT     DX,AL           ;禁止8250内部中断
MOV     DX,3FCH
MOV     AL,0
OUT     DX,AL           ;8250收发方式,禁止中断
RET
I8250   ENDP
CODE    ENDS
END      BEG

```

## 程序2

```

;FILENAME:931_2.ASM
DATA    SEGMENT
TEXT    DB      'THE QUICK BROWN FOX JUMPS OVER LAZY DOG'
        DB      0DH,0AH
LLL     EQU     $-TEXT
ERROR   DB      'COM1 BAD !',0DH,0AH,'$'
DATA    ENDS
CODE    SEGMENT
        ASSUME  CS:CODE,DS:DATA
BEG:    MOV     AX,DATA
        MOV     DS,AX
        CALL    I8250           ;主串口初始化
        MOV     CH,10           ;10行送CH
AGAIN:  MOV     CL,LLL           ;电文长度送CL
        MOV     BX,OFFSET TEXT
SEND:   MOV     AL,[BX]         ;取数据
        MOV     AH,1
        MOV     DX,0
        INT     14H             ;发送数据
        TEST    AH,80H          ;发送成功否 ?
        JNZ     DISPERR         ;失败,转出错处理
RECEIVE: MOV     AH,2
        MOV     DX,0
        INT     14H             ;接收一个数据
        TEST    AH,80H          ;接收成功否 ?
        JNZ     DISPERR         ;失败,转出错处理
        AND     AL,7FH
DISP:   MOV     AH,2
        MOV     DL,AL
        INT     21H             ;送屏幕显示
        INC     BX
        DEC     CL               ;计数
        JNZ     SEND

```

```

                DEC     CH                ;行计数
                JNZ     AGAIN
                JMP     RETURN
DISPERR:        MOV     AH,9
                MOV     DX,OFFSET ERROR
                INT     21H              ;显示出错信息
RETURN:         MOV     AH,4CH
                INT     21H              ;返回 DOS
;-----
I8250          PROC
                MOV     AX,0083H         ;波特率1200
                MOV     DX,0             ;无校验,8位数据
                INT     14H
                RET
I8250          ENDP
CODE           ENDS
END            BEG

```

- 全双工通信测试：编程实现两台机计算机间数据的异步串行发送和接收
  - 甲机和乙机利用主串口进行发送和接收数据。其中发送采用查询方式，接收采用中断方式
  - 数据格式：8个数据位，1个停止位，无校验，通信速率1200波特
  - 甲机和乙机任何一方按下ESC，通信结束，各自返回DOS

```

;FILENAME:9_3.asm
.586
DATA        SEGMENT
            OLD0C    DD    ?
            FLAG     DB    0
DATA        ENDS
CODE        SEMENT USE16
ASSUME CS:CODE,DS:DATA
BEG:        MOV AX,DATA
            MOV DS,AX
            CLI                ;关中断
            CALL I8250         ;主串口初始化
            CALL I8259         ;开放主8259A主串口中断
            CALL RD0C          ;读中断向量
            CALL WR0C          ;写中断向量
            STI                ;开中断
SCANT:      CMP FLAG,-1        ;测试是否收到字符
            JE     RETURN
            MOV DX,3FDH
            IN  AL,DX
            TEST AL,20H
            JZ   SCANT
            MOV AH,1
            INT 16H
            JZ   SCANT
            MOV AH,0
            INT 16H

```

```
ADN AL,7FH
MOV DX,3F8H
OUT DX,AL
CMP AL,1BH
JNZ SCANT
TWAIT: MOV DX,3FDH
IN AL,DX
TEST AL,40H
JZ TWAIT
RETURN: CALL RESET
MOV AH,4CH
INT 21H
```